Assignment 08

Solve the differential equation $\frac{dy}{dx}$ = xy with initial condition y(1)=2 and graph the solution for -2 <= x <= 2.

```
Clear[solu];
solu = DSolve[{y'[x] == x y[x], y[1] == 2}, y[x], x]
Clear[f];
f = solu[[1, 1, 2]]
Plot[f, {x, -2, 2}, PlotStyle → {Blue, Thick},
 AxesLabel → {"x", "y"}, LabelStyle → Directive[Black, Bold, Medium]]
```

$\left\{\left\{y[x] \to 2\, e^{-\frac{1}{2}+\frac{x^2}{2}}\right\}\right\}$

$2\, e^{-\frac{1}{2}+\frac{x^2}{2}}$



According to Newton's law of cooling, the temperature of an object changes at a rate proportional to the difference in temperature between the object and the outside medium. If an object whose temperature is 70^{o}F is placed in a medium whose temperature is 20^{o}F, and is found to be 40^{o}F after 3 minutes, what will its temperature be after 6 minutes?

Do not worry if *Mathematica* complains when you try to find the constant of the equation.

```
Clear[solu];
solu = DSolve[{Temp'[t] == k (Temp[t] - 20), Temp[0] == 70}, Temp[t], t];
Print["Solution of the equation"];
Clear[EqT];
EqT = solu[[1, 1, 2]]

Print["Finding the constant of the equation"];
Clear[ct];
ct = Solve[(EqT /. t → 3) == 40, k];
const = ct[[1, 1, 2]]

Print["Final expression for temperature"];
Clear[FinalEq];
FinalEq = EqT /. k → const

Print["Temperature at 6 min is ", FinalEq /. t → 6, " F."];
```

```
Solution of the equation
```

$10 \times \left(2 + 5\, e^{k\,t}\right)$

```
Finding the constant of the equation
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$-\frac{1}{3}\,\text{Log}\left[\frac{5}{2}\right]$

```
Final expression for temperature
```

$10 \times \left(2 + 2^{t/3} \times 5^{1-\frac{t}{3}}\right)$

```
Temperature at 6 min is 28 F.
```

A baseball is hit with velocity of 100 ft/s at an angle of 30^o with the horizontal.

The height of the bat is 3 ft above the ground. Neglecting air and wind resistance,
(a) will it clear a 35-ft-high fence located 200 ft from home plate?
(b) Make a plot of the trajectory of the baseball. Label the axes.
[Hint: use parametric plot]
(Assume g = 32.16 ft/s^2)

```
Clear[x, y, vo, theta, h, g];
vo = 100.;
theta = 30 Degree;
h = 3.;
g = 32.16;

Clear[solu];
solu = DSolve[{y''[t] == -g, x''[t] == 0, y[0] == h, x[0] == 0,
    y'[0] == vo Sin[theta], x'[0] == vo Cos[theta]}, {x[t], y[t]}, t];

Print["Equations for vertical and horizontal motions"];
Clear[hori, vert];
vert = solu[[1, 1, 2]]
hori = solu[[1, 2, 2]]

Print["Time it takes to reach x = 200 ft"];
Clear[tt, time];
tt = Solve[hori == 200];
time = tt[[1, 1, 2]]

Print["The vertical position when the ball reaches x = 200 ft"];
Clear[yAt200];
yAt200 = vert /. t → time
Print["is smaller than 35 ft, so it won't clear the fence."]

Print[]
Print["Plot of the trajectory"]
ParametricPlot[{hori, vert}, {t, 0, 3.5}, AxesLabel → {"x", "y"}]
```

```
Equations for vertical and horizontal motions
```

$3. + 50.\, t - 16.08\, t^2$

86.6025 t

Time it takes to reach x = 200 ft

2.3094

The vertical position when the ball reaches x = 200 ft

32.7101

is smaller than 35 ft, so it won't clear the fence.

Plot of the trajectory



At what angle should the ball in the previous problem be hit so that it goes over the fence? Give you answer in degrees not radians.
Do not worry about complaints of *Mathematica* about the existence of more solutions. Every angle added by 2Pi is also a solution, this is why it complains.
Discard negative angles.

```
Clear[x, y, vo, theta, h, g];
vo = 100.;
h = 3.;
g = 32.16;

Clear[solu];
solu = DSolve[{y''[t] == -g, x''[t] == 0, y[0] == h, x[0] == 0,
    y'[0] == vo Sin[theta], x'[0] == vo Cos[theta]}, {x[t], y[t]}, t];

Print["Equations for vertical and horizontal motions"];
Clear[hori, vert];
vert = solu[[1, 1, 2]]
hori = solu[[1, 2, 2]]

Print["Time it takes to reach x = 200 ft"];
Clear[tt, time];
tt = Solve[hori == 200];
time = tt[[1, 1, 2]]

Print["The angle to guarantee that the ball reaches y=35ft at x=200ft"];
Clear[ang];
ang = Solve[(vert /. t → time) == 35, theta];

Do[
 If[ang[[k, 1, 2]] > 0, Print["Possible angle = ", ang[[k, 1, 2]] 180 / Pi, " Degree"]];
 , {k, 1, Length[ang]}]
```

```
Equations for vertical and horizontal motions
```

$3. - 16.08\, t^2 + 100.\, t \,\mathrm{Sin}[\mathrm{theta}]$

$100.\, t \,\mathrm{Cos}[\mathrm{theta}]$

```
Time it takes to reach x = 200 ft
```

$2.\,\mathrm{Sec}[\mathrm{theta}]$

```
The angle to guarantee that the ball reaches y=35ft at x=200ft
```

```
Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for
    complete solution information. >>
Possible angle = 30.7838 Degree
Possible angle = 68.3065 Degree
```

The equation governing the amount of current I, flowing through a simple resistance-inductance circuit when an EMF (voltage) is applied is $L \frac{dI}{dt} + R I = E$. The units for E, I, and L are respectively volts, amperes, and henries. If R = 10 ohms, L=1 henry, the EMF source is an alternating voltage whose equation is E(t)=10 sin(5t), and the current is initially 4 amperes, find an expression for the current at time t and plot the graph of the current for the first 3 seconds.

```
Clear[R, L, emf];
R = 10.;
L = 1.;
emf = 10 Sin[5 t];

Clear[sol, curr];
sol = DSolve[{curr'[t] + R curr[t] == emf, curr[0] == 4.}, curr[t], t];

Clear[current];
current = Chop[sol[[1, 1, 2]]]

Plot[current, {t, 0, 3}, PlotStyle → {Thick, Blue}, AxesLabel → {"t", "Current"},
 PlotRange → All, LabelStyle → Directive[Black, Bold, Medium]]
```

$e^{-10. t} (4.4 - 0.4 e^{10. t} Cos[5. t] + 0.8 e^{10. t} Sin[5. t])$



If a spring with mass m attached at one end is suspended from its other end, it will come to rest in an equilibrium position. If the systems is then perturbed by releasing the mass with an initial velocity of vo at a distance yo below its equilibrium position, its motion satisfies the differential equation

$m \frac{d^2 y}{dt^2} + a \frac{dy}{dt} + k\, y = 0,$

y'(0)=vo,

y(0)=yo.

Above, "a" is the damping constant (determined experimentally) due to friction and air resistance, and k is the spring constant given in Hooke's law.

A mass of 1/4 slug is attached to a spring with a spring constant k=6lb/ft. The mass is pulled downward from its equilibrium position 1 ft (that is, yo=-1) and then released. Assuming a damping constant a=1/2, determine the motion of the mass and sketch its graph for the first 5 seconds.

```
Clear[m, vo, yo, k, a, y, t];
m = 1 / 4.;
vo = 0;
yo = -1.;
a = 1 / 2.;
k = 6;

Clear[sol];
sol = DSolve[{m y''[t] + a y'[t] + k y[t] == 0, y'[0] == vo, y[0] == yo}, y[t], t]

Plot[sol[[1, 1, 2]], {t, 0, 5}, PlotStyle → {Thick, Blue},
 AxesLabel → {"t", "Height"}, PlotRange → All,
 LabelStyle → Directive[Black, Bold, Medium]]
{{y[t] → e^(-1. t) (-1. Cos[4.79583 t] - 0.208514 Sin[4.79583 t])}}
```


Height
0.5
−0.5
−1.0
1
2
3
4
5
t

The logistic equation for population growth
$\frac{dp}{dt} = a\, p - b\, p^2$
was discovered in the mid-nineteenth century by the biologist Pierre Verhulst. The constant “b” is generally small in comparison to “a” so that for small population size p, the quadratic term in p will be negligible and the population will grow approximately exponentially. For large p, however, the quadratic term serves to slow down the rate of the growth of the population.
a) Solve the logistic equation for general values of the constants a, b, and initial population po. [Do not worry if *Mathematica* says that more solutions could not be found]
b) Sketch the solution for a=2, b=0.05 and an initial population po=10. [Range of t from 0 to 5].
c) Determine the limiting value of the population at t-> infinity.

```
Clear[a, b, po, p, sol];

Print["Item (a)"]
sol = DSolve[{p'[t] == a p[t] - b p[t]^2, p[0] == po}, p[t], t]

Print[];
Print["Item (b)"]
Plot[sol[[1, 1, 2]] /. {po → 10, a → 2, b → 0.05}, {t, 0, 5},
 PlotStyle → {Thick, Blue}, AxesLabel → {"t", "Population"}, PlotRange → All,
 LabelStyle → Directive[Black, Bold, Medium], AxesOrigin → {0, 0}]

Print[];
Print["Item (b)"]
Limit[sol[[1, 1, 2]] /. {po → 10, a → 2, b → 0.05}, t → Infinity]
```

Item (a)

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{p[t] \to \frac{a\, e^{a\, t}\, po}{a - b\, po + b\, e^{a\, t}\, po}\right\}\right\}$$

Item (b)



Item (b)

40.

Plot the solution to the differential equation

$$\frac{d^2 y}{dt^2} + (\frac{dy}{dt}+1)^2 \frac{dy}{dt} + y = 0$$

y(0)=1

y'(0)=0

for 0<= t <= 10

```
Clear[y, t, sol];
sol = NDSolve[{y''[t] + (y'[t] + 1) ^ 2 × y'[t] + y[t] == 0, y[0] == 1, y'[0] == 0},
   y[t], {t, 0, 10}];

Clear[f];
f = sol[[1, 1, 2]];
Plot[f, {t, 0, 10}, PlotStyle → {Thick, Blue},
 LabelStyle → Directive[Black, Bold, Medium]]
```